
CHIA Enrollment Trends (Data through March 2025)
MA APCD Programming Code

As described in the CHIA Enrollment Trends Technical Appendix, Enrollment Trends monitors unique Massachusetts residents with primary medical insurance coverage using eligibility data from the Massachusetts All Payer Claims Database (MA APCD). MA APCD membership reports are generated on a bi-annual basis and sourced from the most recently submitted payer-specific Member Eligibility (ME) file. Twenty-four months of membership data are used and segmented by product enrollment date spans. 

Provided below are the methodology and programming code used by CHIA to create the MA APCD membership reports for Enrollment Trends. The methodology and code have been discussed with and reviewed by the top 12 commercial payers[footnoteRef:1] (and affiliated entities) but are subject to further development in order to arrive at the most accurate membership counts for each payer. On a case-by-case basis, CHIA may request supplemental data from payers to augment or replace MA APCD enrollment counts.  [1:  ConnectiCare is not included in reporting due to low Massachusetts enrollment.] 


I. Methodology

Enrollment is determined if the 15th day of the last month of each quarter (i.e., Snapshot Date) is within the Product Enrollment Start Date (ME041) and Product Enrollment End Date (ME042) of a member record (i.e., ME041 ≤ Snapshot Date ≤ ME042). Member records with NULL values for ME042 are considered actively enrolled. Members are distinguished using data element CHIACarrierSpecificUniqueMemberID (ME107) by OrgID (ME001). 

MA APCD membership reports do not count duplicate member records. A unique member record is selected according to the following criteria, in order of significance:

De-duplicate by the following order:
1. Medical coverage (ME018=1)
2. Primary insurance (ME028)
a. ME028=1 for primary insurance
b. ME028=2 for non-primary insurance
c. ME028=(3,4,5) for unknown, other, N/A
3. Greatest last activity date (ME056)
4. Greatest Member Eligibility ID (CHIA-assigned record identifier)

Subset to the following:
1. Massachusetts resident (ME016=MA)
2. Medical coverage (ME018=1)
3. Primary insurance (ME028=1)



II. Programming Code[footnoteRef:2] [2:  CHIA is providing this methodology and programming code as a convenience. It has been prepared for informational purposes only and is based on information believed to be reliable. The methodology and programming code are subject to change without notice. CHIA does not provide any guarantee or opinion on its accuracy. CHIA disclaims any liability for the improper or incorrect use of the information contained herein.] 


MA APCD data extraction and aggregation are implemented in SAS, using SQL pass-through to access data in Postgres. The code below may not be inclusive of all data manipulations performed on the final output dataset.

a. SAS/SQL Code


/*************************************************************************************************
	SAS/SQL CODE TO EXTRACT AGGREGATE APCD MEMBERSHIP DATA FOR ENROLLMENT TRENDS REPORTING
*************************************************************************************************/

*--------------- BEGIN ACTION ITEMS -------------------------------------------

	1. Initialize global variables
	2. Perform additional action items below prior to running program

--------------- END ACTION ITEMS ---------------------------------------------;

/* Initialize global variables                                                                    */
%let SYM1= 202406;
%let SYM2= 202409;
%let SYM3= 202503;
%let DataDir= C:\EnrollmentTrend\output;
%let OrgIdList= (290, 291, 295, 296, 300, 301, 302, 312, 3505, 3735, 16621, 4962, 7655, 7789, 8026, 8647, 9913, 10353, 10441, 10444, 10632, 10647, 10728, 10920, 10926, 10929, 20122, 11474, 11726, 12814, 13074, 20741);
%let Snapshot1= '2023-06-15'; 	*First snapshot date, format: 'yyyy-mm-dd';
%let Snapshot2= '2025-03-15'; 	*Last snapshot date, format: 'yyyy-mm-dd';

* Fixed global variables;
libname out "&DataDir.";
%let OutFile= out.ME_ET_&SYM.; 

option compress=binary;


/*************************************************
Begin: Netezza summary data extract
*************************************************/

*--------------- BEGIN ACTION ITEMS -------------------------------------------

	Perform the checks below prior to running code

	1. Review "Processing Summary" below for previous processing info
	2. Check that source data is coming from correct data source

--------------- END ACTION ITEMS ---------------------------------------------;


*	Processing Summary:

	Extract 8 quarters, starting with snapshot date of 12/15/2018, of ME data for Enrollment Trends. The below dataset has the 
	following characteristics:
		1) The dataset has been de-duplicated (to_keep_v1= 1)
		2) The dataset contains only OrgIDs relevant to Enrollment Trends
		3) The dataset contains Product fields: ProductIDNumber, ProductLineofBusinessModel, ProductBenefitType, NaicCode, 
			CarrierLicenseType, ProductName, InsurancePlanMarketCleaned
		4) The dataset contains demographic fields: AgeGroup, County, and MemberGender 

;


proc sql;
	connect to odbc (dsn=zeusgres);
	create table &OutFile. as
	select * from connection to odbc (
	with DateList as (select dates::date from generate_series(&Snapshot1., &Snapshot2., '3 month'::interval) dates
	),
	pr as (select *
    , COALESCE(lag((to_date(submissionyearmonth||'01','yyyymmdd') + INTERVAL '1 MONTH') :: date) 
      over (partition by ORGID, PRODUCTIDNUMBER order by submissionyearmonth),to_date('18000101','yyyymmdd')) eff_start
    , case when lead(submissionyearmonth) 
      over (partition by ORGID, PRODUCTIDNUMBER order by submissionyearmonth) is null 
        then to_date('99991231','yyyymmdd') 
        else (to_date(submissionyearmonth||'01','yyyymmdd') + INTERVAL '1 MONTH'):: date -1 
          end eff_end
        from apcd_centrale.vw_177_pr_consol_ce_hsp
	)
	select OrgID
		, SubmissionControlID
		, SubmissionYearMonth
		, Dates
		, InsuranceTypeCodeProduct
		, MHPlanName
		, MedicalCoverage
		, PrimaryInsuranceIndicator
		, CoverageType
		, MarketCategoryCode
		, SpecialCoverage
		, ProductID
		, ProductLineOfBusinessModel
		, Situs
		, ProductBenefitType
		, NaicCode
		, CarrierLicenseType
		, ProductName
		, InsurancePlanMarketCleaned
		, GIC
		, RACP
		, MemberMA
		, County
		, MemberGender
		, AgeGroup
		, MAExchangeFlag
		, to_keep_v1
/*		, to_keep_v2*/
		, count(*) as nrecs
	from (
	select me.MemberEligibilityID
		, me.OrgID
		, me.SubmissionControlID
		, me.SubmissionYearMonth
		, me.InsuranceTypeCodeProduct
		, MedicalCoverage
		, PrimaryInsuranceIndicator
		, CoverageType
		, me.MarketCategoryCode
		, SpecialCoverage
		, case when me.InsuranceTypeCodeProduct='30' then 
			case when me.OrgID=4962 then
				case when me.ProductIDNumber like '%TogetherCHA%' then 'Tufts Health Together with CHA'
					when me.ProductIDNumber like '%TogetherBCH%' then 'Tufts Health Together with Boston Children’s ACO'
					when me.ProductIDNumber like '%TogetherBIDCO%' then 'Tufts Health Together with BIDCO'
					when me.ProductIDNumber like '%TogetherATRIUS%' then 'Tufts Health Together with Atrius Health'
				end
			when me.OrgID=3505 then
				case when me.ProductIDNumber = 'A001223' then 'WellSense Signature Alliance'
					when me.ProductIDNumber = 'A001221' then 'WellSense Community Alliance'
					when me.ProductIDNumber = 'A001222' then 'WellSense Mercy Alliance'
					when me.ProductIDNumber = 'A001224' then 'WellSense Southcoast Alliance'
					when me.ProductIDNumber = 'A001971' then 'WellSense Beth Isreal Lahey Health Performance Network ACO'
					when me.ProductIDNumber = 'A001972' then 'WellSense Boston Children’s ACO'
					when me.ProductIDNumber = 'A001973' then 'East Boston Neighborhood Health WellSense Alliance'
					when me.ProductIDNumber = 'A001974' then 'WellSense Care Alliance'
				end
			when me.OrgID=296 then
				case when me.ProductIDNumber in ('PBP0006052_RCX','PBP0006055_RCIC','PBP0006058_RCIIM','PBP0006058_RCIM','PBP0006055_RCIM','PBP0006058_RCIC','PBP0006052_RCIX','PBP0006058_RCIIC','PBP0006058_RCIXM') then 'Berkshire Fallon Health Collaborative'
					when me.ProductIDNumber in ('PBP0006056_RCIC','PBP0006053_RCX','PBP0006059_RCIIM','PBP0006059_RCIM','PBP0006053_RCIX','PBP0006059_RCIIC','PBP0006059_RCIC','PBP0006056_RCIM','PBP0006059_RCIXM') then 'Fallon 365 Care'
					when me.ProductIDNumber in ('PBP0006060_RCIIM','PBP0006060_RCIM','PBP0006060_RCIIC','PBP0006057_RCIC','PBP0006054_RCX','PBP0006057_RCIM','PBP0006054_RCIX','PBP0006060_RCIC','PBP0006060_RCIXM') then 'Wellforce Care Plan'
				end
			when me.OrgID=3735 then 'My Care Family'
			when me.OrgID=301 then 'Be Healthy Partnership'
			end 
		end as MHPlanName
		, cast(me.ProductIDNumber as char(20)) as ProductID
		, prd.ProductLineOfBusinessModel
		, prd.Situs
		, prd.ProductBenefitType
		, prd.NaicCode
		, prd.CarrierLicenseType
		, prd.ProductName
		, prd.InsurancePlanMarketCleaned
		, case when APCDIDCode='3' then 'GIC' else 'Non-GIC' end as GIC
		, ProductEnrollmentStartDate as StartDate
		, case when ProductEnrollmentEndDate is null then cast('99991231' as date) else ProductEnrollmentEndDate end as EndDate
		, LastActivityDate
		, CHIACarrierSpecificUniqueMemberID as UniqueMemberID
		, MemberGenderCleaned as MemberGender
	/*	, case when MemberDateofBirth is NULL or MemberDateofBirth in ('', '0', '0.000000') then 'Unknown' */
		, case when extract(year from age(Dates, ad.DT)) >= 0 and extract(year from age(Dates, ad.DT)) < 1 then '< 1' 
			when extract(year from age(Dates, ad.DT)) >= 1 and extract(year from age(Dates, ad.DT)) < 10 then '1-9'
			when extract(year from age(Dates, ad.DT)) >= 10 and extract(year from age(Dates, ad.DT)) < 20 then '10-19'
			when extract(year from age(Dates, ad.DT)) >= 20 and extract(year from age(Dates, ad.DT)) < 27 then '20-26'
			when extract(year from age(Dates, ad.DT)) >= 27 and extract(year from age(Dates, ad.DT)) < 45 then '27-44'
			when extract(year from age(Dates, ad.DT)) >= 45 and extract(year from age(Dates, ad.DT)) < 65 then '45-64'
			when extract(year from age(Dates, ad.DT)) >= 65 and extract(year from age(Dates, ad.DT)) < 75 then '65-74'
			when extract(year from age(Dates, ad.DT)) >= 75 and extract(year from age(Dates, ad.DT)) < 85 then '75-84'
			when extract(year from age(Dates, ad.DT)) >= 85 and extract(year from age(Dates, ad.DT)) < 121 then '85+'
			else 'Unknown' 
			end as AgeGroup
		, RiskAdjustmentCoveredBenefitPlan as RACP
		, case when ma.MemberStateorProvince ='MA' then 1 else 0 end as MemberMA
		, case when ma.MemberStateorProvince ='MA' then MemberCounty else 'Other' end as County
		, PurchasedThroughMassachusettsExchangeFlag as MAExchangeFlag
		, Dates
		, case when lead(me.MemberEligibilityID) over (partition by me.OrgID, CHIACarrierSpecificUniqueMemberID, Dates
			order by case when MedicalCoverage='1' then 1 else 0 end
					, case when PrimaryInsuranceIndicator = '1' then 1
						when PrimaryInsuranceIndicator = '2' then 0
						when PrimaryInsuranceIndicator in ('3','4','5') then -1 
						else -1 end
					, LastActivityDate
					, me.MemberEligibilityID) 
				is null then 1 else 0 
				end as to_keep_v1	/* New De-duplication hierarchy for Enrollment Trends*/
/*		, case when lead(me.MemberEligibilityID) over (partition by me.OrgID, CHIACarrierSpecificUniqueMemberID, Dates
			order by case when MemberStateorProvince='MA' then 1 else 0 end
					, case when MedicalCoverage='1' then 1 else 0 end
					, case when PrimaryInsuranceIndicator in ('1','3','4') then 1 else 0 end
					, case when CoverageType in ('ASO', 'ASW') then 1 
							when CoverageType='UND' then 0
							else -1 end
					, LastActivityDate
					, me.MemberEligibilityID) 
				is null then 1 else 0 
				end as to_keep_v1	/* Old De-duplication hierarchy for Enrollment Trends*/
/*		, case when lead(me.MemberEligibilityID) over (partition by me.OrgID, HashCarrierSpecificUniqueMemberID, Dates*/
/*			order by LastActivityDate*/
/*					, me.MemberEligibilityID) */
/*				is null then 1 else 0 */
/*				end as to_keep_v2	/* De-duplication hierarchy for Premiums */
	  FROM APCD_CENTRALE.VW_177_ME_CONSOL_CE_HSP as me

/* UPDATE isnull */

			inner join DateList 
				on Dates between ProductEnrollmentStartDate and case when ProductEnrollmentEndDate is null then cast('99991231' as date) else ProductEnrollmentEndDate end

/* UPDATE product join criteria */

			left join pr as prd
          		on me.linkorgidpr = prd.orgid and me.ProductIDNumber = prd.ProductIDNumber and Dates between prd.eff_start and prd.eff_end
			left join apcd_centrale.apcd_date as ad 	/* Join to a date table to process only valid Member Date of Birth values */
	 			on MEMBERDATEOFBIRTHYEARMONTH||'01' = ad.DT_YYYYMMDD /*MemberDOBYM only includes year and month of birth, adding dummy day to join to date table.*/
			left join apcd_centrale.tbl_address_me_memberaddress as ma
        		on me.membereligibilityid = ma.membereligibilityid and me.submissioncontrolid = ma.submissioncontrolid and me.submissionyearmonth = ma.submissionyearmonth
	where me.SubmissionYearMonth=&SYM3.
		and me.OrgID in &OrgIdList.
	) a
	where to_keep_v1=1	/*or to_keep_v2=1*/     /* Save out only de-duplicated records */
	group by OrgID
		, SubmissionControlID
		, SubmissionYearMonth
		, Dates
		, InsuranceTypeCodeProduct
		, MHPlanName
		, MedicalCoverage
		, PrimaryInsuranceIndicator
		, CoverageType
		, MarketCategoryCode
		, SpecialCoverage
		, ProductID
		, ProductLineOfBusinessModel
		, Situs
		, ProductBenefitType
		, NaicCode
		, CarrierLicenseType
		, ProductName
		, InsurancePlanMarketCleaned
		, GIC
		, RACP
		, MemberMA
		, County
		, MemberGender
		, AgeGroup
		, MAExchangeFlag
		, to_keep_v1
/*		, to_keep_v2*/
	order by 1,2,3,4
	);
	disconnect from odbc; 
	quit;

/*************************************************
End: Netezza summary data extract
*************************************************/


/*************************************************
Begin: Export summary data extract
*************************************************/

* 	Part 1 of 2: Save out de-duped enrollment counts with ET standard fields 
	The below dataset has the following characteristics:
		1) The dataset has been de-duped (to_keep_v1= 1)
		2) The dataset contains only OrgIDs relevant to Enrollment Trends
		3) The dataset contains two Product fields: ProductID and ProductLineofBusinessModel
		4) The dataset contains demographic fields: AgeGroup, County, and MemberGender 
;

proc sql;
	create table out.ET_&SYM3. as
	select OrgID
		, SubmissionControlID
		, SubmissionYearMonth
		, Dates
		, InsuranceTypeCodeProduct
		, MHPlanName
		, MedicalCoverage
		, PrimaryInsuranceIndicator
		, CoverageType
		, MarketCategoryCode
		, SpecialCoverage
		, ProductID
		, ProductLineOfBusinessModel
		, Situs
		, ProductBenefitType
		, NaicCode
		, CarrierLicenseType
		, ProductName
		, InsurancePlanMarketCleaned
		, GIC
		, RACP
		, MemberMA
		, County
		, MemberGender
		, AgeGroup
		, MAExchangeFlag
		, sum(nrecs) as Member_Count
	from &OutFile.
	where to_keep_v1=1
	group by OrgID
		, SubmissionControlID
		, SubmissionYearMonth
		, Dates
		, InsuranceTypeCodeProduct
		, MHPlanName
		, MedicalCoverage
		, PrimaryInsuranceIndicator
		, CoverageType
		, MarketCategoryCode
		, SpecialCoverage
		, ProductID
		, ProductLineOfBusinessModel
		, Situs
		, ProductBenefitType
		, NaicCode
		, CarrierLicenseType
		, ProductName
		, InsurancePlanMarketCleaned
		, GIC
		, RACP
		, MemberMA
		, County
		, MemberGender
		, AgeGroup
		, MAExchangeFlag
	;
	quit;


* Apply hard-coded groupings and export dataset to Excel 
 	Macro Input: dataEdit(name of input dataset, name of outputdataset) 
 	Macro Output: a temporary SAS dataset and an Excel xlsx-file 

	As of 2/15/17 - Since we are no longer creating "standard" and "demog" Etrends files, we don't need the 'name to identify dataset for Excel filename' as part of the macro.
	It has been removed
;

proc format;
	value OrgName
		290, 7655, 10353, 10441, 10442, 10647, 10929, 20122, 11745 = 'Aetna'
		3735, 16621 = 'MGBHP'
		10632 = 'Anthem'
		291   = 'BCBS'
		3505  = 'WellSense'
		10728 = 'CCA'
		10920 = 'CeltiCare'
		293, 295, 7422, 10447, 11474, 11499, 11726, 11215 = 'CIGNA'
		7041  = 'ConnectiCare'
		296, 8026 = 'Fallon'
		300, 13074   = 'HPHC'
		302   = 'HPI' /* In later processing, re-classify HPI OrgID=302 as CompanyName=HPHC */
		301   = 'HNE'
		3156  = 'MassHealth'
		12226 = 'Minuteman'
		4962  = 'THPP'
		9913  = 'SWH'
		8647  = 'Tufts'  
		312, 313, 7789, 10444, 10925, 10926, 10932, 10933, 10934, 10935, 12814 = 'United'
              20741 = 'Wellcare'
		other = 'Other'
	;

	value $MktSect
		'IND', 'GCV', 'ISCO' = 'Individual'				
		'GS1', 'GS2', 'GS3'  = 'Small Group (1-25)'			
		'GS4'  = 'Small Group (26-50)'		
		'GLG1' = 'Mid-Size Group (51-100)'		
		'GLG2', 'GLG3' = 'Large Group (101-500)'			
		'GLG4' = 'Jumbo-Size Group (501+)'		
		'GSA'  = 'Qualified Association'		
		other  = 'Other' 
	;

	value $CovType
		'UND' = 'Fully-Insured'			
		'ASO', 'ASW' = 'Self-Insured'	
		other = 'Other'
	;

	value $ProdType
		'MC', 'MO', '30' = 'Medicaid'
        'CC' = 'Commonwealth Care'
        'CE' = 'Commonwealth Choice'
        'HM' = 'HMO'
        '12' = 'PPO'
        '13' = 'POS'
        '14' = 'EPO'
        '15' = 'Indemnity'
        '16', 'HN', '20' = 'Medicare'
		'IC' = 'OneCare'
        'SC' = 'Senior Care Options'
         other = 'Other'
	;

	/* Group HPI (OrgID 302) under HPHC */
	value OrgNameReport
		290, 7655, 10353, 10441, 10442, 10647, 10929, 20122, 11745 = 'Aetna'
		10632 = 'Anthem'
		291   = 'BCBS'
		3505  = 'WellSense'
		10728 = 'CCA'
		10920 = 'CeltiCare'
		293, 295, 7422, 10447, 11474, 11499, 11726, 11215 = 'CIGNA'
		7041  = 'ConnectiCare'
		296, 8026 = 'Fallon'
		300, 302, 13074  = 'HPHC'
		301   = 'HNE'
		3156  = 'MassHealth'
		12226 = 'Minuteman'
		3735, 16621  = 'MGBHP'
		9913  = 'SWH'
		8647  = 'Tufts'  
		4962  = 'THPP'
		312, 313, 7789, 10444, 10925, 10926, 10932, 10933, 10934, 10935, 12814 = 'United'
              20741 = 'Wellcare' 
		other = 'Other'
	;
	run;

	
%macro dataEdit(dataInfile, dataOutfile);
PROC SQL;
	CREATE TABLE &dataOutfile. AS
	SELECT *
		/*11/19/18 LM: HPHC used to submit Medicare Supplement policies with InsTypeCodeProduct 15 (see Data Finding 23).
		This population is being submitted as InsTypeCodeProduct=SP as of 201607 ME submission.*/
		/*, CASE WHEN ORGID=300 AND INSURANCETYPECODEPRODUCT='15' THEN '2'
		/*11/19/18 LM: Per Data Anomaly Tracker, Aetna submits PrimaryInsuranceIndicator='3' for members who haven't filed any claims. See note
		at top of page indicating that we changed to PrimaryCoverage='2' as of 5/31/17.
			when ORGID in (290, 7655, 10353, 10441, 10442, 10647, 10929, 11745) and PRIMARYINSURANCEINDICATOR='3' then '2'
				ELSE PRIMARYINSURANCEINDICATOR
				END AS PRIMARYCOVERAGE*/
		, case when primaryinsuranceindicator='1' then '1'
			else primaryinsuranceindicator end as PRIMARYCOVERAGE
		, put(MarketCategoryCode, $MktSect.)as GROUPSIZE
		/*  - Data Finding #2: Fallon (OrgID=8026) incorrectly classified its self-insured population as
				InsuranceTypeCodeProduct='09'. As a result, cannot break out its self-
				insured population by product type. All other members are fully-insured.
			- Resolution: Need to hardcode MARKET='Self-Insured' for InsuranceTypeCodeProduct='09'
				and MARKET='Fully-Insured' for all other members for ET reporting. 
			- Removed this modification 11/20/18 LM*/	
		, CASE 
				/*  Finding: United (OrgID=7789) is a student health plan, which is classified as 
						CoverageType='STN'.
					Resolutions: Need to hardcode MARKET='Fully-Insured' for ET reporting. */	
				WHEN ORGID=7789 THEN
					CASE WHEN CoverageType in ('STN', 'UND') THEN 'Fully-Insured'	
				  		WHEN CoverageType in ('ASO', 'ASW') THEN 'Self-Insured'	
				  		ELSE 'Other' END
				ELSE put(CoverageType, $CovType.) end as MARKET
		/* 	- Finding: Network Health's (OrgID=4962) MSP population is classified as Product Type='MO'. 
					- Resolution: Need to hardcode PLANTYPE='Medical Security Program' for Product Type='MO' for ET reporting.
						Also have to hardcode OneCare population until they begin submitting using InsTypeCodeProduct = IC
						- Removed this logic for 201903. THPP started submitting IC for One Care, and MSP is no longer relevant.
					- Removed 'CI'-->'HMO' logic. 'CI' only appeared in 201412 and 201503 submissions. 12/10/18 LM*/	
	     /*, CASE WHEN ORGID=4962 THEN 				
						CASE WHEN InsuranceTypeCodeProduct='MO' THEN 'Medical Security Program'
                       	 	WHEN InsuranceTypeCodeProduct='MC' THEN CASE WHEN MarketCategoryCode= 'ISCO' THEN 'OneCare' ELSE 'Medicaid' END
							WHEN InsuranceTypeCodeProduct='30' THEN 'Medicaid'
                        	WHEN InsuranceTypeCodeProduct='CC' OR SpecialCoverage='CC' THEN 'Commonwealth Care'
                        	WHEN InsuranceTypeCodeProduct='CE' THEN 'Commonwealth Choice'
                        	WHEN InsuranceTypeCodeProduct='HM' THEN 'HMO'
                        	WHEN InsuranceTypeCodeProduct='12' THEN 'PPO'
                        	WHEN InsuranceTypeCodeProduct='13' THEN 'POS'
                        	WHEN InsuranceTypeCodeProduct='14' THEN 'EPO'
                        	WHEN InsuranceTypeCodeProduct='15' THEN 'Indemnity'
                        	WHEN InsuranceTypeCodeProduct in ('16', 'HN', '20') THEN 'Medicare'
                        	WHEN InsuranceTypeCodeProduct='SC' THEN 'Senior Care Options'
							WHEN InsuranceTypeCodeProduct='IC' THEN 'OneCare'
                  		ELSE 'Other' END
           			WHEN SpecialCoverage='CC' THEN 'Commonwealth Care'*/
			/*11/20/18 LM: Added custom logic to make sure Fallon ZZ records go under SCO Plan Type as opposed to Other since we classify Other as Commercial*/
			, CASE WHEN OrgID=8026 and InsuranceTypeCodeProduct='ZZ' and MarketCategoryCode='ISCO' THEN 'Senior Care Options'
                ELSE put(InsuranceTypeCodeProduct, $ProdType.)
				END AS PLANTYPE
		, put(OrgId, OrgName.) as CompanyName
		, put(OrgID, OrgNameReport.) as CompanyName_Report
		, CASE WHEN UPCASE(County) in ('BARNSTABLE', 'BERKSHIRE', 'BRISTOL', 'DUKES', 'ESSEX',
					'FRANKLIN', 'HAMPDEN', 'HAMPSHIRE', 'MIDDLESEX', 'NANTUCKET', 'NORFOLK', 
					'PLYMOUTH', 'SUFFOLK', 'WORCESTER', '') THEN County
				ELSE 'Other'
				END AS MEMBERCOUNTY
	FROM &dataInfile.;
	QUIT;

%mend dataEdit;
%dataEdit(out.ET_&SYM3., out.report_&SYM3.);

/*************************************************
Begin: Export summary data extract
*************************************************/
/* Export data into txt-file and 1 SAS dataset to pivot in Power Pivot */

/* PROCESSING TIME: */
proc datasets library=work nolist;
    delete temp;
quit;

data work.temp(drop=Dates);
      set out.report_&SYM3. (obs=max);
      Date = Dates;
      format Date MMDDYYS10.;
run;

proc export data=work.temp
      outfile="&DataDir3.\ET_&SYM3..txt"
      dbms=dlm replace;
      delimiter='|';
run;
/*************************************************
End: Export summary data extract

